

# USBメモリ 点・MBKデータの読み書き (MPC-3000)

ACCEL ti2k-181114 (LastUpdate 11/14/18)

## ■ 点データ、MBKデータの書き込み、読み込み

- ファイルへ書き込み : USB\_PSAVE data count filename (追加書き込み。新規の場合は事前に USB\_DEL)
- ファイルから読み込み : USB\_PLOAD filename (事前にMPCの使用範囲をクリア)
- ファイル削除 : USB\_DEL filename
- 各コマンド 正常終了のとき EOF(0)==1, ファイル無し EOF(0)==2
- MPC-3000(SH2A) BL/I 2.00\_09 2018/11/08

## • プログラム例

```
IF USB(0) <> 1 THEN /* USBメモリ有無確認
  PR "USB Memory None"
  END
END_IF
GOSUB *ClearData /* 使用範囲のデータ消去
PR "Create Data"
FOR p=0 TO 9 /* データ作成
  SETP p+1 p+1 p+2 p+3 p+4 : MBK(p+100)=p+100
NEXT
PR "Delete File"
USB_DEL "PNTDATA.P2K" /* ファイル消去
USB_DEL "MBKDATA.MBK"
PR "SAVE to Memory"
USB_PSAVE P(1) 10 "PNTDATA.P2K" /* 点データ保存
USB_PSAVE MBK(100) 10 "MBKDATA.MBK" /* MBKデータ保存
GOSUB *ClearData /* 使用範囲のデータ消去
PR "LOAD from Memory"
USB_PLOAD "PNTDATA.P2K" /* 点データ読み込み
IF EOF(0)==2 THEN /* ファイルが無いと EOF(0)==2
  PR "=File None="
ELSE
  FOR p=1 TO 10 : PR "P(" p ")=" P(p) : NEXT /* 確認表示
END_IF
USB_PLOAD "MBKDATA.MBK" /* MBKデータ読み込み
IF EOF(0)==2 THEN
  PR "=File None="
ELSE
  FOR p=100 TO 109 : PR "MBK(" p ")=" MBK(p) : NEXT /* 確認表示
END_IF
PR "Completed"
END
```

```
*ClearData
PR "Clear Data"
FILL P(1) 10
FILL MBK(100) 10
RETURN
```

## • 実行結果

```
# Clear Data
Create Data
Delete File
SAVE to Memory
Clear Data
LOAD from Memory
P( 1 )= 1 2 3 4
P( 2 )= 2 3 4 5
(中略)
P( 9 )= 9 10 11 12
P( 10 )= 10 11 12 13
MBK( 100 )= 100
MBK( 101 )= 101
(中略)
MBK( 108 )= 108
MBK( 109 )= 109
Completed
```

```
#DIR ←ダイレクトコマンド ファイル一覧
PNTDATA.P2K 2018/11/14 15:42:36 171
MBKDATA.MBK 2018/11/14 15:42:36 150
#
```

- パソコンのエディタで開いたファイル。
- テキストデータです。

"PNTDATA.P2K"

```
SETP 1 1 2 3 4
SETP 2 2 3 4 5
SETP 3 3 4 5 6
SETP 4 4 5 6 7
SETP 5 5 6 7 8
SETP 6 6 7 8 9
SETP 7 7 8 9 10
SETP 8 8 9 10 11
SETP 9 9 10 11 12
SETP 10 10 11 12 13
```

"MBKDATA.MBK"

```
S_MBK 100 100
S_MBK 101 101
S_MBK 102 102
S_MBK 103 103
S_MBK 104 104
S_MBK 105 105
S_MBK 106 106
S_MBK 107 107
S_MBK 108 108
S_MBK 109 109
```

■ MS-EXCELで作成したカンマ区切りCSVから点データに変換するサンプル

- USB\_READ a\$ コマンドで1行ずつ読み込み SETP コマンドで点データとして登録します。
- ファイル名は 予約変数 FILE\$ で指定。読んだ文字列は文字列変数 a\$ に入ります。
- 先頭の文字が数値で無い場合は登録しません。

- EXCELで作成したCSVファイルの内容

"test.csv"

```
1, 1, 28, 29578, -10000    ← 点番号, X, Y, U, Z の順
2, 1, 113, 18813, -10000
3, 1, 170, 18819, -10000
4, 4, 5, 6, 7
```

"test1.csv"

```
1, 1111, 1111, 1111, 1111
2, 2222, 2222, 2222, 2222    ←EXCELでは空行。先頭は数字ではない。
, , , ,
3, 3333, 3333, 3333, 3333
4, 4444, 4444, 4444, 444
```

• プログラム例

```
FILE$="test.csv"          /* Read File Name
PR FILE$

FILL P(1) 2000           /* P(1)~2000point Clear
GOSUB *CheckPoint

IF USB(0)<>1 THEN        /* Check USB Memory
  PR "USB Memory None"
  END
END_IF

DO
  USB_READ a$           /* Read a line
  IF EOF(0)==3 THEN     /* Check USB Mem Exist
    BREAK
  END_IF
  IF EOF(0)==2 THEN     /* Check File Exist
    PRINT "File None"
    BREAK
  END_IF
  IF EOF(0)==1 THEN     /* Check End Of File
    PRINT "End Of File"
    BREAK
  END_IF
  PR a$
  p=VAL(a$)             /* if not numeric then VOID
  IF p<>VOID THEN
    x=VAL(0) : y=VAL(0) : u=VAL(0) : z=VAL(0)
    PR "SETP" p x y u z
    SETP p x y u z
  END_IF
LOOP
USB_READ -1            /* Seek Pointer Reset

GOSUB *CheckPoint

END

*CheckPoint
FOR p=1 TO 4
  PR "P(" p ")=" P(p)
NEXT
RETURN
```

• 実行結果

```
RUN
# test.csv
P( 1 )= 0 0 0 0
P( 2 )= 0 0 0 0
P( 3 )= 0 0 0 0
P( 4 )= 0 0 0 0
1, 1, 28, 29578, -10000
SETP 1 1 28 29578 -10000
2, 1, 113, 18813, -10000
SETP 2 1 113 18813 -10000
3, 1, 170, 18819, -10000
SETP 3 1 170 18819 -10000
4, 4, 5, 6, 7
SETP 4 4 5 6 7
End Of File
P( 1 )= 1 28 29578 -10000
P( 2 )= 1 113 18813 -10000
P( 3 )= 1 170 18819 -10000
P( 4 )= 4 5 6 7
```

```
RUN
# test1.csv
P( 1 )= 0 0 0 0
P( 2 )= 0 0 0 0
P( 3 )= 0 0 0 0
P( 4 )= 0 0 0 0
1, 1111, 1111, 1111, 1111
SETP 1 1111 1111 1111 1111
2, 2222, 2222, 2222, 2222
SETP 2 2222 2222 2222 2222
, , , ,
3, 3333, 3333, 3333, 3333
SETP 3 3333 3333 3333 3333
4, 4444, 4444, 4444, 444
SETP 4 4444 4444 4444 444
End Of File
P( 1 )= 1111 1111 1111 1111
P( 2 )= 2222 2222 2222 2222
P( 3 )= 3333 3333 3333 3333
P( 4 )= 4444 4444 4444 444
```